THE UNIVERSITY OF ALABAMA AT BIRMINGHAM

CS330
C to Assembly

Spring 2022

Lab 8

Or Higher Order Language Appreciation Week

Helpful Hints

- Assembly is just like procedural programming in other languages, but the building blocks are smaller, and there are a lot less helper tools
  - The process is the same - Break the problem down into smaller pieces
  - Use the tools in our toolkit to solve the problem, build up the tools we need
- It's a manual drive, nothing is automatic -- We need to keep a lot of program state info in our head, or better, write it down, e.g. what's in memory, and where. Or "where did I leave my variable?"
  - A quick sketch of the Registers and Stack can be extremely helpful
- The computer is a shared space (shared resources), following the rules / contract is key to harmony
  - Register management: pass/receive args from functions in certain registers, caller and callee save responsibilities
  - Stack management: leave it like you found it, check stack alignment
- Code documentation is key. It's not uncommon to have more comment lines than code.
  - `# single line comment`
  - `/* block comment */`
- We'll use x86 AT&T syntax in this course
  - Be careful, there is a lot of Intel syntax (and other syntax, and bad info) on the interwebs

High Level (just for today) Computer Architecture



Registers

- 16 General Purpose Registers
- Register names per AT&T syntax
- Will not use floating, vector registers in this course
- Can also access subsets



Register	Usage	Old Names	Args	Saved by	Preserved Across Function Calls
%rax	temporary register; with variable arguments passes information about the number of vector registers used; **1st return register**	accumulator		**Caller**	No
%rbx	callee-saved register; optionally used as base pointer	base		**Callee**	Yes
%rcx	used to pass 4th integer argument to functions	counter, loop counter	4	**Caller**	No
%rdx	used to pass 3rd argument to functions; **2nd return register**	data	3	**Caller**	No
%rsp	stack pointer	stack pointer		**Callee**	Yes
%rbp	callee-aved register, optionally used as frame pointer	base pointer		**Callee**	Yes
%rsi	used to pass 2nd argument to functions	source index	**2**	**Caller**	No
%rdi	used to pass 1st argument to functions	destination index	**1**	**Caller**	No
%r8	used to pass 5th argument to functions		5	**Caller**	No
%r9	used to pass 6th argument to functions		6	**Caller**	No
%r10	temporary register, used for passing a function's static chain pointer			**Caller**	No
%r11	temporary register			**Caller**	No
%r12 - r15	callee-saved registers			**Callee**	Yes

GNU Assembler (AS) Manual: https://sourceware.org/binutils/docs/as/index.html#SEC_Contents

eflags
(and how to view registers)

- In GDB
  (i)nfo (r)egisters eflags

```
(gdb) info registers eflags
eflags         0x202    [ IF ]
```

- To show all general purpose registers, including %rip (instruction pointer), eflags
  (i)nfo (r)egisters all

  or individually via
  (i)nfo (r)egisters $<name>
  e.g. (i)nfo (r)egisters $rax

  or
  tui reg general



How to read / interpret the syntax

- Typical AT&T mnemonics use three letter instructions with a one letter suffix to represent the size

Suffix		
b	byte	1 byte
w	word	2 bytes
l	doubleword	4 bytes
q	quadword	8 bytes



Instruction		Effect	Description	pg
Data Movement				
mov	S, D	$D \leftarrow S$	Move source to destination (movslq, sign extend l to q, pg 222)	183
push	S	$R[\%rsp] \leftarrow R[\%rsp] - 8$ $M[R[\%rsp]] \leftarrow S$	push source onto stack	189
pop	D	$D \leftarrow M[R[\%rsp]]$ $R[\%rsp] \leftarrow R[\%rsp] + 8$	pop top of stack into destination	189
Arithmetic				
lea	S, D	$D \leftarrow \&S$	load effective address	191
add	S, D	$D \leftarrow D + S$	add	192
sub	S, D	$D \leftarrow D - S$	subtract	192
mul	S, D	$D \leftarrow D * S$	multiply	192
imulq	S	R[%rdx]:R[%rax] $\leftarrow S * R[\%rax]$	multiply (2 64 bit numbers)	198
xor	S, D	$D \leftarrow D \hat{} S$	exclusive-or	192
cqto		R[%rdx]:R[%rax] $\leftarrow$ SignExtend($R[\%rax]$)		
idivq	S	$R[\%rdx]$ $\leftarrow$ R[%rdx]:R[%rax] mod S $R[\%rax]$ $\leftarrow$ R[%rdx]:R[%rax] / S	signed divide	198
Control				
cmp	S_1, S_2	S_2 - S_1	compare	202
jmp	label		direct jump	205
jmp	*Operand		indirect jump	205
je	label		jump if equal / zero (Zero Flag set)	205

S = Source, D = Destination

Operands take one of these three forms

1. Immediate / Literal: $4
2. Register: %rax
3. Memory

Type	From	Operand Value	Name
Immediate	$Imm	Imm	Immediate
Register	r_a	R[r_a]	Register
Memory	Imm	M[Imm]	Absolute
Memory	(r_a)	M[R[r_a]]	Indirect
Memory	Imm(r_b)	M[Imm + R[r_b]]	Base + displacement
Memory	Imm(r_b, r_i, s)	M[Imm + R[r_b] + (R[r_i] * s)]	Scaled Indexed

(see Book, pg 181 for more)

- Imm refers to a constant value, e.g. 0x8048d8e, 48
- r_a refers to a register
- R[r_a] refers to the value stored in register r_a
- M[x] refers to the value stored at memory address x

Note: can't move (mov) from Memory to Memory

- multiple levels of understanding
  1. The Syntax, what does the specific word mean? e.g movl
  2. The Semantics, what does the word mean in context, ie. what does the sentence mean? E.g. $\mathrm{movl\ \$0,\ \%eax}$
     what does the paragraph mean?



THE UNIVERSITY OF ALABAMA AT BIRMINGHAM

Compile C to Assembly

C to Assembly



- Using **-S** will run the compiler, generating an assembly file *add.s,* but will NOT call the assembler, and so will not produce an executable file.
- Instead of our usual gcc command, run this instead:

gcc add.c –S

Or better (turn off optimization): **gcc add.c –S –O0**

- If you have done this correctly, you should now see a new file, *add.s*

add.c

lab08 > C add.c > main()

```c
#include<stdio.h>

int main(){
    int a, b, c;
    printf("Enter an integer: ");
    scanf("%d", &a);
     printf("Enter a second integer: ");
    scanf("%d", &b);

    c = a + b;

    printf("Sum of %d and %d is %d \n", a, b, c);

    return 0;
}
```

lab08 > ASM add.s

```
1     .file   "add.c"
2     .text
3     .section    .rodata
4   .LC0:
5     .string "Enter an integer: "
6   .LC1:
7     .string "%d"
8   .LC2:
9     .string "Enter a second integer: "
10  .LC3:
11    .string "Sum of %d and %d is %d \n"
12    .text
13    .globl  main
14    .type   main, @function
15  main:
16  .LFB0:
17    .cfi_startproc
18    pushq   %rbp
19    .cfi_def_cfa_offset 16
20    .cfi_offset 6, -16
21    movq    %rsp, %rbp
22    .cfi_def_cfa_register 6
23    subq    $32, %rsp
24    movq    %fs:40, %rax
25    movq    %rax, -8(%rbp)
26    xorl    %eax, %eax
27    leaq    .LC0(%rip), %rdi
```

```
28    movl    $0, %eax
29    call    printf@PLT
30    leaq    -20(%rbp), %rax
31    movq    %rax, %rsi
32    leaq    .LC1(%rip), %rdi
33    movl    $0, %eax
34    call    __isoc99_scanf@PLT
35    leaq    .LC2(%rip), %rdi
36    movl    $0, %eax
37    call    printf@PLT
38    leaq    -16(%rbp), %rax
39    movq    %rax, %rsi
40    leaq    .LC1(%rip), %rdi
41    movl    $0, %eax
42    call    __isoc99_scanf@PLT
43    movl    -20(%rbp), %edx
44    movl    -16(%rbp), %eax
45    addl    %edx, %eax
46    movl    %eax, -12(%rbp)
47    movl    -16(%rbp), %edx
48    movl    -20(%rbp), %eax
49    movl    -12(%rbp), %ecx
50    movl    %eax, %esi
51    leaq    .LC3(%rip), %rdi
```

add.s

```
52    movl    $0, %eax
53    call    printf@PLT
54    movl    $0, %eax
55    movq    -8(%rbp), %rcx
56    xorq    %fs:40, %rcx
57    je  .L3
58    call    __stack_chk_fail@PLT
59  .L3:
60    leave
61    .cfi_def_cfa 7, 8
62    ret
63    .cfi_endproc
64  .LFE0:
65    .size   main, .-main
66    .ident  "GCC: (Ubuntu 9.1.0-2ubuntu2~18.04) 9.1.0"
67    .section    .note.GNU-stack,"",@progbits
68
```

State of the machine when we receive it

Registers	
%rax	
%rcx	
%rdx	
%rsi	
%rdi	
%rsp	
%rbp	



Preamble

Registers	
%rax	
%rcx	
%rdx	
%rsi	
%rdi	
%rsp	
%rbp	%rsp




```
15  main:
16  .LFB0:
17      .cfi_startproc
18      pushq   %rbp
19      .cfi_def_cfa_offset 16
20      .cfi_offset 6, -16
21      movq    %rsp, %rbp
```

Prep for local variables

```
4        int a, b, c;
```

Registers	
%rax	
%rcx	
%rdx	
%rsi	
%rdi	
%rsp	
%rbp	%rsp

```
23       subq    $32, %rsp
```



Setup Stack Canary

(Vulcan gcc doesn't include)

Registers	
%rax	~~%fs:40~~ 0
%rcx	
%rdx	
%rsi	
%rdi	
%rsp	
%rbp	%rsp

```
24      movq    %fs:40, %rax
25      movq    %rax, -8(%rbp)
26      xorl    %eax, %eax
```



Let's print

```
5       printf("Enter an integer: ");
```

Registers	
%rax	0 ~~0~~ 18
%rcx	
%rdx	
%rsi	
%rdi	Pointer to .LCO "Enter an integer: " **(maybe)**
%rsp	
%rbp	%rsp

```
27      leaq    .LC0(%rip), %rdi
28      movl    $0, %eax
29      call    printf@PLT
```



Let's get some user input, int a

aka:

```
6       scanf("%d", &a);
```

Registers	
%rax	Pointer to %rbp - 20 ~~0~~ 1
%rcx	
%rdx	
%rsi	Pointer to %rbp - 20
%rdi	Pointer to .LC1 "%d"
%rsp	
%rbp	%rsp

```
31      leaq    -20(%rbp), %rax
32      movq    %rax, %rsi
33      leaq    .LC1(%rip), %rdi
34      movl    $0, %eax
35      call    __isoc99_scanf@PLT
```



Let's get some user input, int b aka:

```
8       scanf("%d", &b);
```

Registers	
%rax	Pointer to %rbp - 16 ~~0~~ 1
%rcx	
%rdx	
%rsi	Pointer to %rbp - 16
%rdi	Pointer to .LC1 "%d"
%rsp	
%rbp	%rsp

```
40      leaq    -16(%rbp), %rax
41      movq    %rax, %rsi
42      leaq    .LC1(%rip), %rdi
43      movl    $0, %eax
44      call    __isoc99_scanf@PLT
```



Let's add

```
10      c = a + b;
```

Registers	
%rax	~~b~~ a + b
%rcx	
%rdx	a
%rsi	
%rdi	
%rsp	
%rbp	%rsp

```
45      movl    -20(%rbp), %edx
46      movl    -16(%rbp), %eax
47      addl    %edx, %eax
48      movl    %eax, -12(%rbp)
```



Print

```
12      printf("Sum of %d and %d is %d \n", a, b, c);
```

Registers	
%rax	~~a~~ ~~0~~ 20
%rcx	c
%rdx	b
%rsi	a
%rdi	Pointer to .LC3 “Sum of %d and ….”
%rsp	
%rbp	%rsp

```
49      movl    -16(%rbp), %edx
50      movl    -20(%rbp), %eax
51      movl    -12(%rbp), %ecx
52      movl    %eax, %esi
53      leaq    .LC3(%rip), %rdi
54      movl    $0, %eax
55      call    printf@PLT
```



Clean-up and return

```
14      return 0;
```

Registers	
%rax	0
%rcx	~~Stack Canary (%fs:40)~~ 0
%rdx	
%rsi	
%rdi	
%rsp	
%rbp	%rsp




```
56      movl    $0, %eax
57      movq    -8(%rbp), %rcx
58      xorq    %fs:40, %rcx
59      je  .L3
60      call    __stack_chk_fail@PLT
61  .L3:
62      leave
63      .cfi_def_cfa 7, 8
64      ret
```



Clean-up and return

```
14      return 0;
```

Registers	
%rax	0
%rcx	0
%rdx	
%rsi	
%rdi	
%rsp	
%rbp	Old %rbp

```
56      movl    $0, %eax
57      movq    -8(%rbp), %rcx
58      xorq    %fs:40, %rcx
59      je  .L3
60      call    __stack_chk_fail@PLT
61  .L3:
62      leave
63      .cfi_def_cfa 7, 8
64      ret
```